
11.1.2016
DFCM Energy Program Director’s Preface
This document is meant to give the State’s Agencies and Institution’s building operators, commissioning agents, and other facilities management professionals an additional set of tools to assist in them their efforts to proactively operate their facilities.  While DFCM does not endorse Skyspark, SkyFoundry, or Relevant Solutions it does endorse the use of leveraging building analytical data to better understand and inform the day to day operations of a building.  While Skyspark is the basis of this document there are many other software platforms that provide building analytical data in a similar fashion.  Each has its merits and should be considered in the context of the goals of each facilities management group.  As with any tool it is only valuable when it is used and used properly.

In 2014 I first introduced Skyspark to SLCC by coordinating a presentation for several state agencies by a local business that I had learned had cost effectively implemented this building analytics platform.  Over the next several months SLCC spent time performing due diligence and eventually implemented the platform on several of their local campuses.  In doing so they became well versed in implementing Skyspark.  The following is a joint effort prepared by SLCC and Relevant Solutions to disseminate those lessons learned including providing the basic “rules” to be used by government building professionals in their effort to reduce their total cost of ownership.

Intent
This document contains basic information on how to install and run SkySpark. It also has some basic rules to implement for your systems. It is a supplemental to SkyFoundry’s website on how to install and run. There will be links to their webpage on how to carry out the actual steps. 

Setup
SkySpark is a great tool for analytics. The way that SLCC has set it up on their system is on its own virtual machine. It stores all its data in its own SQL database and is able to be queried by its internal Folio app. SLCC has a server for their BAS system with its own database of trends. They also have a virtual machine for their utility monitoring system and dashboard. All three of these systems are separate and together make up their energy information system. SkySpark is the parent to these other two systems by monitoring and reporting of things not going as designed or intended. When issues are found, it informs the user and the work begins on correction to the systems.

The reason there is three separate systems are for redundancy of data and to keep each item doing what it does best. BAS loves doing its own controls of the HVAC system with light trending. The Energy system loves tracking utility usage and with a dashboard displaying the information and having its own database. Skyspark loves to track and trend the other two systems on its own database and running analytics on that database.

You will need to purchase a license for SkySpark from Relevant Solutions or a third party company who is installing it for you. The license is a flat rate that is relatively inexpensive. You will also need to buy caps. These are pretty much a number of points you are going to want to trend in Skyspark. For instance, if you have 1 building with 1-2 AHU’s and minor other equipment, you will probably need around 500 caps. SLCC has purchased around 100,000 caps and has used about 50,000 for 15 buildings with VAV systems and 3 central plants etc. There is an annual renewal of the license which cost is based on how many caps you own. If you choose not to renew, you won’t have access to the weather data or program updates, but the rest of the system remains yours and will remain active. You could renew anytime and update the system later if you choose. Renewal is fairly inexpensive, at approximately $2-4k for 100,000 caps. Once you have your license you are ready to install.

Install
A great resource on how to install SkySpark is found here: https://skyfoundry.com/file/151/Installing-and-Updating-SkySpark.pdf

Operational Parameters – Rules and Tagging
The following is a list of common building analytics rules. The intention behind these rules is to bring attention to the systems that are not working correctly. It is not to identify every possible problem in the system, just to let the user know something is not correct, and to notify them to check into it. These are simple high-level rules. We believe that rules that are too specific will cause excessive alarms and will eventually get ignored.

The main set up of rules are basic SQL calls that will inquire based off the tags assigned to the rule. When these tags are identified, it will run some function calls while using data from these systems.  In other words, a piece of equipment has attributes such as MA or RA. We will mark this piece with these tags. We will operate a SQL function call that will find all the points with the MA and RA tag and send it to a function. This function will then analyze the data values such as length of time and temperature to see if it violates what is being tested. When this happens, it will create a message, known as a spark, to inform the user that this piece of equipment is in violation of the rule.

To describe the set up of these rules, we will have a name for the rule, the purpose of the rule and the tagging needed for the rule. We will then display the function programming for that rule with its name.

1. Name of rule
a. Description of the rule.
b. Tags for the equipment the rule will look at.
c. What the rule will do and test.
d. Name of the function.
e. Programming for the function.
There is a supplemental document that contains the tagging convention for the majority of HVAC equipment and meters. When bringing in points into SkySpark, these tags will need to be added. The points will also need to have its history activated based on the recommended amount such as COV or interval.

AHU Systems

1. AHU Mixed Air Temperature (MA Temp) Abnormal.
a. This rule will look at all AHU’s that have the points MA temp and RA temp 
b. The proper tag(s): temp, sensor, mixed, air, and return
c. Compare MA temp value to RA Temp, if MA Temp is above RA temp then spark.
If MA temp is below 50 F then spark 
d. Name is ahuMatAbnormal
e. //Relevant 2016 - Trevor Adelman
(ahu, dates) => do
  //reads mixed air temp sensor for given AHU (MAT with tags mixed, air, temp, outside, and sensor)
  mat: read(mixed and air and temp and outside and sensor and equipRef==ahu->id)
  
  //read return air temp sensor for given AHU (RAT with tags return, air, temp, sensor and not outside so that it doesnt pull the mixed air temp)
  rat: read("return" and air and temp and sensor and not outside and equipRef==ahu->id)
  
  //fan status point
  fan: read(discharge and air and fan and status and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x => x)
  
  //joins the histories and interpolates so that we can compare the two
  joined: hisRead([mat, rat], dates).hisInterpolate()
  
  //maps the joined histories to find the difference between the MAT and RAT
  comp: joined.map r => do
          {ts: r->ts, diff: r["v0"] - r["v1"]}
  end
  
  //finds periods where MAT > RAT
  spark1: comp.hisFindPeriods(v => v > 5)
  
  //find periods where MAT is less than 50 (using same tags as above)
  spark2: mat.hisRead(dates).hisFindPeriods(v=> v < 50).hisFindAll(v => v > 1)

  //find union of both the above spark parameters
  sparkUnion: hisPeriodUnion([spark1,spark2])

  sparkFinal: hisPeriodIntersection([sparkUnion, fan])

end

2. AHU Cool and Econ mode
a. Look at all equip’s that have the points cool valve and damper . 
b. tag(s): cmd, valve, cool, outside, damper, cmd, unit and not minVal
c. Compare the cool valve periods that are active to the econ damper when its active at the same time.  
d. Name of func is ahuCoolandEconMode
e. //Relevant 2016 - Trevor Adelman
(ahu, dates, minTime: 0min) => do
// normalize minTime to hours
minTime = minTime.to(1h)

// get periods when cooling and heating
cool: read(cool and cmd and valve and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x => x > 10)
// 
econ: readAll(outside and damper and cmd and unit and not minVal and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x => x > 10)
// compute intersection of those periods
hisPeriodIntersection([cool, econ])
.findAll(r => r->v0 >= minTime)
end

3. Discharge Air Temp (DA Temp) Abnormal – 
a. Look at all equip’s that have the points DA temp, DA temp sp, and DA fan status. 
b. tag(s):  temp, discharge, sensor, air, sp, fan, and status.
c. Compare DA temp to DA temp sp when DA fan status is on, if it differs by more than 2 F   for more than 10 minutes then spark.  
d. Name of func is ahuDatAbnormal
e. //Relevant 2016 - Trevor Adelman

(ahu, dates) => do
  
  //reads DAT setpoint
  datSP: read(discharge and temp and sp and equipRef==ahu->id)
  
  //reads DAT
  dat: read(discharge and temp and sensor and not sp and equipRef==ahu->id)
  
  //reads fan command point
  fan: read(discharge and air and fan and status and equipRef==ahu->id)
  
  //joins DAT and DAT setpoint into same history
  joined: hisRead([datSP, dat], dates).hisInterpolate()
  
  //maps the two points finding the difference between the two and taking the absolute value of the difference
  comp: joined.map r => do
          {ts: r->ts, diff: abs(r["v0"] - r["v1"])}
  end
  
  //finds all periods that fan is on
  spark1: fan.hisRead(dates).hisFindPeriods(v=>v)
  
  //finds all periods where the difference between DAT and the DAT setpoint is more than +/-3 degrees difference for more than 25 min
  spark2: comp.hisFindPeriods(v=> v > 3).hisFindAll(x => x > 25min)
  
  //finds intersection of 2 previous spark parameters so that it only sparks when both are happening at the same time
  sparkFinal: hisPeriodIntersection([spark1,spark2])

  end

4. Simultaneous Heating and Cooling –
a. Look at all equip’s that have the points CHW valve, HW valve, and Preheat valve.
b. tag(s):  chilled, water, valve, heating, and glycol
c. If CHW valve has a current value of greater than 0%, while HW valve or Preheat valve also have current value of greater than 0% then spark.
d. Name of func is ahuCoolandHeat
e. //Relevant 2016 - Trevor Adelman

(ahu, dates, minTime: 0min) => do

// normalize minTime to hours
  minTime = minTime.to(1h)

  status: read(point and fan and status and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x=>x)

  // get periods when cooling and heating
  cwv: read(chilled and water and valve and cmd and cool and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x => x > 5)

  hwv: read(hot and water and heating and cmd and valve and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x=> x > 5)

  pre: read(heating and hot and pre and water and valve and cmd and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x=> x > 5)
  
  // compute union of hot water valve and preheat valve
  union: hisPeriodUnion([hwv, pre])
    
  //compute intersection of unionized (heat or preheat) and chilled water valve
  spark: hisPeriodIntersection([cwv, union])
 
  finalSpark: hisPeriodIntersection([spark, status])
 
 end

5. AHU Duct Static Pressure Reset Malfunctioning – 
a. Look at all AHU’s that have the point(s) DA pressure sp.
b. tag(s):  discharge, air, pressure, and sp.  
c. If current value doesn’t change within an 8-hour period, then spark.
d. Name of func is ahuDuctPressureReset
e. //Relevant 2016 - Trevor Adelman

(ahu, dates) => do
  
  //Reads in DA pressure setpoint, reads the date, finds the range of every minute of data, 
  //finds where the range is zero, and returns the periods where the value doesn't change 
  //from zero for more than 8 hours at a time. 
  read(discharge and air and pressure and sp and equipRef==ahu->id)
        .hisRead(dates)
        .hisRollup(spread, 1hr)
        .hisFindPeriods(v => v == 0)
        .hisFindAll( x => x > 8hr)
  end

6. AHU DA temp Set point reset Malfunctioning –
a. Look at all AHU’s that have the point(s) DA temp sp.
b. tag(s):  discharge, air, temp, sp
c. If current value doesn’t change over an 8 hour period, then spark.
d. Name of func is ahuahuDatSetpointReset
e. //Relevant 2016 - Trevor Adelman

(ahu, dates) => do
  
  //Reads in DA pressure setpoint, reads the date, finds the range of every minute of data, 
  //finds where the range is zero, and returns the periods where the value doesn't change 
  //from zero for more than 8 hours at a time. 
  spark: read(discharge and air and temp and sp and equipRef==ahu->id)
            .hisRead(dates)
            .hisRollup(spread, 1hr)
            .hisFindPeriods(v => v == 0)
            .hisFindAll( x => x > 8hr)
  
  status: read(point and fan and status and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x=>x)

  finalSpark: hisPeriodIntersection([spark, status])
    
end

7. AHU Fan Speed High 
a. Look at all AHU’s that have the point(s) DA Fan speed 
b. tag(s): discharge, air, fan, speed
c. If current value stays at >95% for more than 8 hours straight then spark
d. Name of func is ahuFanSpeedHigh
e. //Relevant 2016 - Trevor Adelman

(ahu, dates) => do
  
  //Reads in DA fan speed, reads the date, finds the range of every minute of data, 
  //finds where the range is zero, and returns the periods where the value doesn't change 
  //from zero for more than 8 hours at a time. 
 spark: readAll(discharge and air and fan and speed and equipRef==ahu->id)
                .hisRead(dates)
                .hisFindPeriods(v => v > 95)
                .hisFindAll( x => x > 8hr)

  //status: read(point and fan and status and equipRef==ahu->id).hisRead(dates).hisFindPeriods(x=>x)
  
  //finalSpark: hisPeriodIntersection([spark, status])

end

8. AHU On and Fan Off
a. Look at all equip’s that have the fan status and command. 
b. tag(s): fan, status, cmd.
c. Compare the fan status to its command, it they don’t match up, spark.
d. Name of func is ahuOnAndFanOffNew
e. (ahu, dates, minTime: 0min) => do
  // normalize minTime to hours
  minTime = minTime.to(1h)

  // get our periods 
  fanOff: read(fan and status and equipRef==ahu->id)

  onPeriods: read(fan and cmd and quipRef==ahu->id)

  // compute intersection of periods
//  hisPeriodIntersection([fanOff, hisPeriodUnion(onPeriods)])
//     .findAll(r => r->v0 >= minTime)

end

DUAL DUCT VAV System

9. DD VAV Cool and Heat
a. Look at all equipment that have the points cooling and heating periods from other functions below. You will need to also add the other functions listed in this section for these to work as they are dependent on each other. 
b. tag(s):dualDuct.
c. Compare the periods when cooling damper and heating damper are open at the same time then spark.  
d. Name of func is ddVavCoolAndHeat
e. (dualDuct, dates, minTime: 0min) => do
  // normalize minTime to hours
  minTime = minTime.to(1h)

  // get periods when cooling and heating
  cool: ddVavCoolPeriods(dualDuct, dates)
  heat: ddVavHeatPeriods(dualDuct, dates)

  // compute intersection of those periods
  hisPeriodIntersection([cool, heat])
    .findAll(r => r->v0 >= minTime)
 end


10. DD VAV Cool Periods
a. Look at all equip’s that have the points dualDuct. 
b. tag(s):  dualDuct.
c. Compare This function only calculates the cooling periods on a dualDuct box.  
d. Name of func is ddVavCoolPeriods
e. (dualDuct, dates, match:true) => do
  dualDuct.toDdVavCool.hisRead(dates).hisFindPeriods(matchPointVal(_, match))
end

11. DD VAV Heat Periods
a. Look at all equip’s that have the points dualDuct. 
b. tag(s):  dualDuct.
c. Compare This function only calculates the heating periods on a dualDuct box.  
d. Name of func is ddVavHeatPeriods
e. (dualDuct, dates, match:true) => do
  dualDuct.toDdVavHeat.hisRead(dates).hisFindPeriods(matchPointVal(_, match))
end

12. DD VAV Cool
a. Look at all equip’s that have the points dualDuct. 
b. tag(s):  dualDuct, coldDeck, cmd.
c. Compare This function only identifies the dual duct boxes that have cooling dampers. 
d. Name of func is toDdVavCool 
e. (dualDuct, checked:true) => do
  cools: readAll(coldDeck and cmd and equipRef == dualDuct->id)

  if (cools.size > 1) cools = cools.findAll(p => p["stage"] == 1)
  if (cools.isEmpty) do
    if (checked) throw "No 'coldDeck' point defined: " + dis(dualDuct)
    return null
  end
  cools.first
end

13. DD VAV Heat
a. Look at all equip’s that have the points dualDuct. 
b. tag(s):  dualDuct, hotDeck, cmd.
c. Compare This function only identifies the dual duct boxes that have heating dampers.
d. Name of func is toDdVavHeat  
e. (dualDuct, checked:true) => do
  heats: readAll(hotDeck and cmd and equipRef == dualDuct->id)

  if (heats.size > 1) heats = heats.findAll(p => p["stage"] == 1)
  if (heats.isEmpty) do
    if (checked) throw "No 'hotDeck' point defined: " + dis(dualDuct)
    return null
  end
  heats.first
end

Standard VAV with Reheat Systems

14. VAV not sized correctly –
a. Look at all VAV’s that have the point(s) DA flow and CMAX flow
b. tag(s):  discharge, air, flow, cfm, sensor, max, sp, and cooling.
c. If current status of DA flow = current status of CMAX flow for more than 8 hours, spark
d. Name of func is vavNotSizedCorrectly
e. //Relevant 2016 - Trevor Adelman

(vav, dates) => do
 
   //find Discharge Air Flow trend
   daFlow: read(discharge and air and flow and sensor and cfm and equipRef==vav->id).hisRead(dates)
 
   //finds cMax trending
   cMax: read(cMax and cfm and sp and equipRef==vav->id).hisRead(dates)
   
   //joins the two histories
   joined: hisJoin([daFlow,cMax]).hisInterpolate()
   
   //maps the histories together so that the result is the difference at every timestamp
   compMap: joined.map r => do
          {ts: r->ts, diff: r["v0"] - r["v1"]}
  end 
   
   //sparks when there is no difference between the two histories (both points are the same) for 8 or more hours straight
   spark: compMap.hisFindPeriods(v=> v == 0).hisFindPeriods( x => x > 8hr)
 
end

15. Hot Water Valve malfunctioning –
a. Look at all equip’s that have the point(s) HW valve, and DA temp.
b. tag(s): heating, hot, water, valve, cmd, sensor, discharge, air, temp
c. If current value of HW valve is >35% then look at DA temp and if it is < 85 degrees spark
d. Name of func is vavHotWaterValveMalfunction
e. //Relevant 2016 - Trevor Adelman

(vav, dates) => do
  
  //finds hot water valve reading for given vav
  hwv: read(hot and water and heating and valve and cmd and equipRef==vav->id).hisRead(dates)
  
  //finds discharge air temp for given vav
  daTemp: read(discharge and air and temp and sensor and equipRef==vav->id).hisRead(dates)
  
  //finds periods of time where the hot water valve has a reading above 35
  spark1: hwv.hisFindPeriods(v => v > 35)
  
  //finds periods of time where the discharge air temp is above 85
  spark2: daTemp.hisFindPeriods(x => x < 85)
  
  //finds periods of time where both spark parameters above are met
  finalSpark: hisPeriodIntersection([spark1, spark2])
  
end

16. Occupancy sensor malfunctioning – 
a. Look at all equip’s that have the point(s) Occupancy
b. tag(s): occupancy, sensor, zone  
c. If current status is on for > 8 hours then spark.
d. Name of func is vavOccupancySensorMalfunction
e. //Relevant 2016 - Trevor Adelman

(vav, dates) => do

    //reads in occupancy point for given VAV, reads in the history, finds periods where 
    //the value is true, and only shows periods above the threshold of 8 hours

    read(occupy and equipRef==vav->id)
          .hisRead(dates)
          .hisFindPeriods(v => v)
          .hisFindPeriods(x => x > 8)
  
end

17. VAV Insufficient flow  
a. Look at all VAV’s that have the point(s) DA damper
b. tag(s): discharge, air, cmd, damper
c. If current value of DA damper is >95% for >8 hours then spark.
d. Name of func is vavInsufficientFlow
e. //Relevant 2016 - Trevor Adelman

(vav, dates) => do
  
  //finds discharge air damper position for a given vav, reads the 
  //history, finds periods where value exceeds 95%, and only returns
  //periods that exceed 8 hours
  
  read(discharge and air and damper and cmd and equipRef==vav->id)
          .hisRead(dates)
          .hisFindPeriods(v => v > 95)
          .hisFindPeriods(x => x > 8)
 
end

18. VAV Sizing Cool
a. Looks at
b. Tag(s):
c. If
d. Name is vavSizingCool
e. (equip, dates) => do
  
  read(flow and (equipRef==equip->id)).hisRead(dates).hisFindPeriods(v => v > (read(cmax and (equipRef==equip->id))->curVal)).hisFindAll( v=> v> 8hr)  
  
end

19. VAV Sizing Heat
a. Looks at
b. Tag(s):
c. If
d. Name is vavSizingHeat
e. (equip, dates) => do
    
  read(flow and (equipRef==equip->id)).hisRead(dates).hisFindPeriods(v => v > (read(hmax and (equipRef==equip->id))->curVal)).hisFindAll( v=> v> 8hr)
    
end

Standard FCU

20. FCU cannot adequately heat or cool – 
a. Look at all FCU’s that have the points CHW valve and HW valve
b. tag(s):  chilled, water, valve, heating, and cmd.
c. If either CHW valve or HW valve current value is 100% for >4 hours then spark
d. Name of func is fcuHeatCoolProblem
e. //Relevant 2016 - Trevor Adelman

(fcu, dates) => do

  //reads history for chilled water valve point for given FCU equip
  cwv: read(chilled and water and valve and cmd and cool and equipRef==fcu->id).hisRead(dates)
  
  //reads history for hot water valve point for given FCU equip
  hwv: read(hot and water and heating and valve and cmd and equipRef==fcu->id).hisRead(dates)
  
  //finds periods where above CWV reads 100% for over 4 hours
  spark1: cwv.hisFindPeriods(v => v == 100).hisFindPeriods( x => x > 4)
  
  //finds periods where above HWV reads 100% for over 4 hours
  spark2: hwv.hisFindPeriods(v => v == 100).hisFindPeriods( x => x > 4)
  
  //finds the union of both above spark parameters
  sparkFinal: hisPeriodUnion([cwv, hwv])

end

21. Pump speed High –
a. Look at all equipment that has the point(s) pump speed.
b. tag(s): pump, speed, sensor
c. If current status is >95% for > 4 hours then spark
d. Name of func is pumpSpeedHigh
e. //Relevant 2016 - Trevor Adelman

(point, dates) => do

point.hisRead(dates).hisFindPeriods(x => x > 95).hisFindAll( y => y > 4)

end

22. Cold Pump not tracking correctly  - 
a. Look at all equipment that has the point(s) CHW differential pressure, HW differential pressure, CHW differential sp, and HW differential sp
b. tag(s): chilled, heating, water, differential, pressure, sp 
c. If current value of CHW/HW differential pressure is > or < CHW/HW differential pressure sp by at least 1 psi for > 2 hours then spark. 
d. Name of func is pumpNotTrackingCold
e. (equip, dates) => do
 cwd: read(point and chilled and water and differential and not sp and equipRef==equip->id).hisRead(dates)
cwsp: read(point and sp and chilled and water and differential and equipRef==equip->id).hisRead(dates)
 joined: hisJoin([cwd, cwsp]).hisInterpolate()
mapped: joined.map row => do
{
ts: row->ts,
val: abs(row.get("v0") - row.get("v1"))
}
end
spark1: mapped.hisFindPeriods(x => x > 1)
/////////////////////////////////////////////////////////////////////////////////////////////////////
//Status parameter//
status: readAll(point and status and equipRef==equip->id)
// list of points
points: status.toRecList
// list of history period grids
periods: points.map pt => hisRead(pt, dates).hisFindPeriods(x=>x)
spark2:hisPeriodUnion(periods)
finalSpark: hisPeriodIntersection([spark1,spark2]).hisFindAll(t => t >2)
 end

23. Hot Pump not tracking correctly  - 
a. Look at all equipment that has the point(s) CHW differential pressure, HW differential pressure, CHW differential sp, and HW differential sp
b. tag(s): chilled, heating, water, differential, pressure, sp 
c. If current value of CHW/HW differential pressure is > or < CHW/HW differential pressure sp by at least 1 psi for > 2 hours then spark. 
d. Name of func is pumpNotTrackingCold
e. (equip, dates) => do
/////////////////////////////////////////////////////////////////////////////////////////////////////
//Pump Parameter//
 hwd: read(point and water and differential and not sp and equipRef==equip->id).hisRead(dates)
 hwsp: read(point and sp and differential and equipRef==equip->id).hisRead(dates) joined: hisJoin([hwd, hwsp]).hisInterpolate()
mapped: joined.map row => do
{
ts: row->ts,
val: abs(row.get("v0") - row.get("v1"))
}
end
spark1: mapped.hisFindPeriods(x => x > 1).hisFindAll(t => t >2)
/////////////////////////////////////////////////////////////////////////////////////////////////////
//Status parameter//
status: readAll(point and status and equipRef==equip->id)
// list of points
points: status.toRecList
// list of history period grids
periods: points.map pt => hisRead(pt, dates).hisFindPeriods(x=>x)
spark2:hisPeriodUnion(periods)
finalSpark: hisPeriodIntersection([spark1,spark2])
end

Cooling Towers

24. Cooling tower speed high –
a. Look at all Cooling Towers that have the point(s) Fan speed.
b. tag(s):  fan, sensor, and speed.
c. if current value is >95% for longer than 8 hours then spark.
d. Name of func is coolingTowerSpeedHigh
e. //Relevant 2016 - Trevor Adelman

(equip, dates) => do

 sparkPoint: read(point and equipRef->id==equip->id and fan and speed)
                     .hisRead(dates)
                     .hisFindPeriods(x => x > 95)
                     .hisFindAll(y => y > 8)
end

25. Cooling Tower malfunctioning –
a. Look at all Cooling Towers that have the point(s) CW return temp, CW supply temp, and Fan speed 
b. tag(s): condenser, water, return, sensor, temp, supply, fan speed 
c. Compare CW return temp current value with CW supply temp current value and if it isn’t 5 F > the CW supply temp current value and the current value of Fan speed is > 80% then spark.
d. Name of func is coolingTowerMalfunction
e. //Relevant 2016 - Trevor Adelman

(ct, dates) => do
  
  //chilled water return point
  cwReturn: read(point and water and "return" and condenser and temp and equipRef==ct->id).hisRead(dates)
  
  //chilled water supply point
  cwSupply: read(point and water and supply and condenser and temp and equipRef==ct->id).hisRead(dates)
  
  //cooling tower fan speed
  fanSpeed: read(point and fan and speed and equipRef==ct->id).hisRead(dates)
  
  //joined supply and return
  joined: hisJoin([cwReturn, cwSupply]).hisInterpolate()
  
  //mapped for difference between supply and return
  mapped: joined.map row => do
            {
            ts: row->ts,
            val: abs(row.get("v0") - row.get("v1"))
            }
          end

  //Finds periods of 5 degrees difference or more between supply and return
  spark1: mapped.hisFindPeriods(x => x < 5)
  
  //finds periods where fan speed is above 80
  spark2: fanSpeed.hisFindPeriods(v => v > 80)
  
  //finds intersection of two spark parameters
  sparkFinal: hisPeriodIntersection([spark1, spark2])

end

26. Cooling Tower Sump Heater malfunctioning –
a. Look at all Cooling Towers that have the point(s) Sump heater and OSA temp
b. tag(s) condenser, water, sump, heater, outside, air, temp
c. If current value of OSA temp is >40F and Sump Heater current value is on then spark 
d. Name of func is coolingTowerSumpHeater
e. //Relevant 2016 - Trevor Adelman

(equip, dates) => do

sump: read(point and sump and heater and equipRef->id==equip).hisRead(dates).hisFindPeriods(x => x)

osa: read(point and outside and air and temp and equipRef->id==equip).hisRead(dates).hisFindPeriods(x => x > 40)

spark: hisPeriodIntersection(sump, osa)

end

27. Cooling Tower Freeze protection- 
a. Look at all cooling towers that have the point(s) Sump temp
b. tag(s) condenser, water, sump, temp
c. If Sump temp current value is below 36 F then spark.
d. Name of func is coolingTowerFreeze
e. //Relevant 2016 - Trevor Adelman

(ct, dates) => do

  //finds sump temp point and returns periods where the value is less than 36 degrees
  read(point and sump and water and temp and condenser and equipRef==ct->id)
          .hisRead(dates)
          .hisFindPeriods(x => x < 36)

end

Boilers

28. Boilers short cycling –
a. Look at 
b. tag(s) sensor, status
c. If boiler status current value changes more than once in a 30 minute period then spark 
d. Name of func is boilerShortCycle
e. //Relevant 2016 - Trevor Adelman

(boiler, dates) => do

  boilerStatusOn: read(point and sensor and status and equipRef==boiler->id)
                      .hisRead(dates)
                      .hisFindPeriods( x => x )
                      .hisFindAll( y =>  y < 30min )
  
  boilerStatusOff: read(point and sensor and status and equipRef==boiler->id)
                      .hisRead(dates)
                      .hisFindPeriods( x => not x )
                      .hisFindAll( y =>  y < 30min )

  hisPeriodUnion([boilerStatusOn, boilerStatusOff])

end

29. Boiler malfunction –
a. Look at all Boilers that have the point(s) status, return temp, and supply temp.
b. tag(s): sensor, status, supply, return, temp.
c. When boiler status current value is “on” for > 10 minutes, compare return temp current value to supply temp current value, if it is not > 5 degrees then spark
d. Name of func is boilerMalfunction
e. //Relevant 2016 - Trevor Adelman

(boiler, dates) => do

  //finds boiler status point
  bStatus: read(point and status and not pump and equipRef==boiler->id).hisRead(dates)
  
  //finds return temp point
  rTemp: read(point and temp and entering and equipRef==boiler->id).hisRead(dates)
  
  //finds supply temp point
  sTemp: read(point and temp and supply and not sp and equipRef==boiler->id). hisRead(dates)

  //joins supply and return
  joined: hisJoin([rTemp, sTemp]).hisInterpolate()
  
  //finds difference in supply and return
  mapped: joined.map row => do
            {
            ts: row->ts,
            val: abs(row.get("v0") - row.get("v1"))
            }
          end
  
  //finds periods where difference is less than 5 degrees
  spark1: mapped.hisFindPeriods(x => x<5)
  
  //finds periods when boiler status is on
  spark2: bStatus.hisFindPeriods(x => x).hisFindPeriods(y => y > 10min)
  
  //sparks when both parameters are met
  sparkFinal: hisPeriodIntersection([spark1, spark2])

end

30. Boiler Supply Temp reset malfunction –
a. Look at all boilers that have the point(s) status and supply temp.
b. tag(s): sensor, status, supply, temp
c. If boiler status current value is on for > 8 hours and boiler supply temp current value hasn’t changed during that time then spark.
d. Name of func is boilerSupplyTempMalfunction
e. (boiler, dates) => do

  status: read(point and status and equipRef==boiler->id).hisRead(dates).hisFindPeriods(r => r).hisFindAll(t => t > 8hr)

  sTemp: read(supply and temp and equipRef==boiler->id)
            .hisRead(dates)
            .hisRollup(spread, 1hr)
            .hisFindPeriods(v => v == 0)
            .hisFindAll( x => x > 8hr)
            
  hisPeriodIntersection([status, sTemp])

end

Chillers

31. Chillers short cycling –
a. Look at all Chillers that have the point(s) status
b. tag(s) sensor, status
c. If chiller status current value changes more than once in a 30 minute period then spark 
d. Name of func is chillerShortCycling
e. //Relevant 2016 - Trevor Adelman

(chiller, dates) => do

// read(point and sensor and status and hisSize and equipRef==chiller->id)
//       .shortCycling(dates ,minOffTime: 30min, minOnTime: 30min)

status: readAll(point and sensor and status and hisSize and equipRef==chiller->id)

 // list of points
points: status.toRecList

// list of history period grids
periods: points.map pt => pt.shortCycling(dates ,minOffTime: 30min, minOnTime: 30min)

spark2:hisPeriodUnion(periods)

end

32. Chiller malfunction – 
a. Look at all chillers that have the point(s) status, return temp, and supply temp.
b. tag(s): sensor, status, supply, return, temp.
c. When chiller status current value is “on” for > 10 minutes, compare return temp current value to supply temp current value, if it is not > 5 degrees then spark.
d. Name of func is chillerMalfunction
e. //Relevant 2016 - Trevor Adelman

(equip, dates) => do

  chStatus: read(point and status and equipRef==equip->id).hisRead(dates).hisFindPeriods(x => x).hisFindAll(y => y > 10min)
  
  chSupply: read(point and chilled and supply and temp and equipRef==equip->id).hisRead(dates)
  
  chReturn: read(point and chilled and "return" and temp and equipRef==equip->id).hisRead(dates)

  joined: hisJoin([chReturn, chSupply]).hisInterpolate()
  
  //mapped for difference between supply and return
  mapped: joined.map row => do
            {
            ts: row->ts,
            val: abs(row.get("v0") - row.get("v1"))
            }
          end

  //Finds periods of 5 degrees difference or more between supply and return
  spark1: mapped.hisFindPeriods(x => x > 5)
  
  spark2: chStatus

  hisPeriodIntersection([spark1, spark2])
   
end

33. Chiller Supply Temp reset malfunction –
a. Look at all chillers that have the point(s) status and supply temp.
b. tag(s): sensor, status, supply, temp
c. If chiller status current value is on for > 8 hours and chiller supply temp current value hasn’t changed during that time then spark.
d. Name of func is chillerSupplyTempMalfunction
e. (chiller, dates) => do

  status: read(point and status and equipRef==chiller->id).hisRead(dates).hisFindPeriods(r => r).hisFindAll(t => t > 8hr)

  sTemp: read(entering and temp and equipRef==chiller->id)
            .hisRead(dates)
            .hisRollup(spread, 1hr)
            .hisFindPeriods(v => v == 0)
            .hisFindAll( x => x > 8hr)
            
  hisPeriodIntersection([status, sTemp])

end

General Tag Equipment

34. Equipment Status and enable Mismatch – 
a. Look at all equipment statuses and commands
b. tag(s): cmdRef
c. Compare current status of equipment to the current status of the cmd
d. Name of func is equipStatusCheck
e. //Relevant 2016 - Trevor Adelman

(myPoint,dates) => do

  statusOn: myPoint.hisRead(dates).hisFindPeriods(x => x)
  
  cmdOff: read(point and id==myPoint->id)->cmdRef
  
  cmdOffRead: cmdOff.hisRead(dates).hisFindPeriods(x => not x)
  
  spark1: hisPeriodIntersection([statusOn, cmdOffRead])


  statusOff: myPoint.hisRead(dates).hisFindPeriods(x => not x)
  
  cmdOn: read(point and id==myPoint->id)->cmdRef
  
  cmdOnRead: cmdOff.hisRead(dates).hisFindPeriods(x => x)
  
  spark2: hisPeriodIntersection([statusOff, cmdOnRead])

  hisPeriodUnion([spark1, spark2])

end

35. Equipment Status and occupancy Mismatch- 
a. Look at all equipment statuses
b. tag(s): equip, status, schedRef
c. Compare current status of equipment to the current status of the associated occupancy schedule and spark if they mismatch.
d. Name of func is equipOccupyMismatch
e. //Relevant 2016 - Trevor Adelman

(myStatus, dates) => do

  statusOn: myStatus.hisRead(dates).hisFindPeriods(x => x)
  
  occOff: read(point and occupied and equipRef==myStatus->equipRef).hisRead(dates).hisFindPeriods(x => not x)
   
  spark1: hisPeriodIntersection([statusOn, occOff])

  statusOff: myStatus.hisRead(dates).hisFindPeriods(x => not x)
  
  occOn: read(point and occupied and equipRef==myStatus->equipRef).hisRead(dates).hisFindPeriods(x => x)
   
  spark2: hisPeriodIntersection([statusOff, occOn])

  hisPeriodUnion([spark1, soark2])

end

36. DHW Supply temperature below set point for more than 10 minutes. 
a. Look at domestic HW temperature sensors and compare them to their set point. 
b. tag(s):  Hot, water, system, temp, sensor, sp.
c. If current status of HW temperature sensor doesn’t match the current status of the HW temperature sensor for more than 15 minutes, then spark.
d. Name of func is dhwSupplyTemp
e. //Relevant 2016 - Trevor Adelman

(equip, dates) => do
  
  dhwTemp: read(domestic and hot and water and supply and temp and sensor and not sp and equipRef==equip->id).hisRead(dates)
  
  dhwSP: read(domestic and hot and water and supply and temp and sp and equipRef==equip->id).hisRead(dates)

  joined: hisJoin([dhwTemp, dhwSP]).hisInterpolate()
  
  mapped: joined.map row => do
            {
            ts: row->ts,
            val: abs(row.get("v0") - row.get("v1"))
            }
          end

  spark: mapped.hisFindPeriods(x => x > 0).hisFindAll(y => y >= 10min)

end
image1.jpeg
The Depar of Adm

D1v1910r1 of

Facilities Constru
& Management

SERVICES ELEVATED





image2.png
It Lake &G
cont Lake 59,
College




image3.png
relevant




